

UNLIMITED

2



AD-A221 459


RSRE
MEMORANDUM No. 4335

ROYAL SIGNALS & RADAR ESTABLISHMENT

SHORT RANGE, CLOSE TO GROUND, VHF/UHF PROPAGATION:
A COMPARISON OF MEASURED RESULTS WITH A
SIMPLE RAY-BASED MODEL

Author: P R Bellamy

PROCUREMENT EXECUTIVE,
MINISTRY OF DEFENCE,
RSRE MALVERN,
WORCS.



RSRE MEMORANDUM No. 4335

UNLIMITED

Best Available Copy

0089087

CONDITIONS OF RELEASE

DI-112454

.....

U

COPYRIGHT (c)
1968
CONTROLLER
HMSO LONDON

.....

Y

Reports quoted are not necessarily available to members of the public or to commercial organisations.

ROYAL SIGNALS AND RADAR ESTABLISHMENT

Memorandum 4335

TITLE: SHORT RANGE, CLOSE TO GROUND, VHF/UHF PROPAGATION:
A COMPARISON OF MEASURED RESULTS WITH A SIMPLE RAY-BASED MODEL.

AUTHOR: P. R. Bellamy

DATE: October 1989

ABSTRACT

This report describes a VHF/UHF propagation prediction model and a series of field measurements aimed at verifying the results of the model. Comparisons between the model and the trial results were made at spot frequencies from 100 MHz to 900 MHz for several types of antenna and for various heights (not exceeding 2 metres) above ground level.

The predicted path losses from the model were found to agree quite well with the results of the field measurements. It should be noted that the prediction model was never intended to be a comprehensive tool, merely a utility package which could be incorporated into other software.

The program was written on an HP 9000 series model 310 computer using BASIC 5.1. This language has a set of commands which allow direct manipulation of complex values, consequently, translation into another language or version of BASIC not having these facilities will require extra programming.



Accession For	
NTIS GRA&I	☒
DTIC TAB	☒
Unannounced	☐
Justification	
By	
Distribution/	
Availability Codes	
Dist	Availability/
	Special
A-1	

Copyright
C
Controller HMSO London
1989

CONTENTS

Abstract

1. Introduction	1
2. Theoretical background to the prediction model	1
2.1 Ground reflection coefficient	1
2.2 Calculation of path loss	2
3. Propagation prediction software	3
4. Measurement of propagation loss	4
4.1 Introduction	4
4.2 Description of equipment employed	4
4.3 Description of trial site	5
5. Results of the propagation measurement trial	7
6. Discussion of trial results	7
7. Conclusions	8
8. Acknowledgements	8
9. References	8
Appendix 1 Results of propagation loss measurements	9
Appendix 1 Propagation prediction software	29
Appendix 3 Trial control software	34
Appendix 4 Ground reflection coefficient plots	44

SHORT RANGE, CLOSE TO GROUND, VHF/UHF PROPAGATION:
A COMPARISON OF MEASURED RESULTS WITH A SIMPLE RAY-BASED MODEL.

1. Introduction.

A requirement exists to study the propagation of signals between 100 MHz and 1GHz at ranges of less than 50m and with antenna heights less than 2m. A propagation loss prediction program was written employing geometric ray-theory which agreed quite closely with previously published data (Ref.1). However, this data did not include results obtained when one or both of the antennas is much less than 1m above the ground. It was therefore necessary to undertake a series of measurements using various types of antenna to validate predictions obtained from the model under the conditions for which no data was available.

This report describes both the propagation model and the trials devised to validate the predicted results.

2. Theoretical background to the prediction model.

Energy radiated from a transmitting antenna may be considered to consist of three parts - a direct wave which travels along a direct path from transmitting antenna to receiving antenna, an earth-reflected wave which arrives at the receiving antenna following a reflection at the earth's surface, and a surface wave which is supported by and travels along the earth's surface. The direct wave and the ground reflected wave (the sum of which is known as the space wave) can be accounted for by ray theory whereas the surface wave requires a solution of suitable electromagnetic field equations.

The propagation model calculates propagation loss between two similar antennas by evaluating the sum of the direct ray and the ground reflected ray at the receiving antenna. The effect of the surface wave is relatively small at frequencies above about 100 MHz and is omitted from the model (it does become more significant if the antennas heights are small in terms of wavelengths and if vertical rather than horizontal polarisation is used but in the present context the error is still small).

2.1 Ground reflection coefficient.

The electrical properties of the ground are represented by the following equation for refractive index (assuming $\epsilon_r = 1$, $\sigma = 0$, for air),

$$n = \sqrt{(\epsilon_r - j \frac{\sigma}{\omega \epsilon_0})}$$

where ϵ_r = ground relative permittivity

σ = ground conductivity

$\omega = 2\pi f$

ϵ_0 = permittivity of free space (8.854×10^{-12} F/m)

μ_r is taken as unity for both ground and air.

Typical values for the frequency range under consideration are:

For wet soil, $\epsilon_r = 15$ $\sigma = 0.1$

For dry soil, $\epsilon_r = 4$ $\sigma = 0.0001$

The following equations give the ground reflection coefficient in terms of n^2 (Ref.3) :

For vertical polarisation,

$$R = \frac{n^2 \sin \theta - \sqrt{(n^2 - \cos^2 \theta)}}{n^2 \sin \theta + \sqrt{(n^2 - \cos^2 \theta)}}$$

For horizontal polarisation,

$$R = \frac{\sin \theta - \sqrt{(n^2 - \cos^2 \theta)}}{\sin \theta + \sqrt{(n^2 - \cos^2 \theta)}}$$

where n^2 is defined above and θ is the angle of incidence between the ray and the horizontal.

If these functions are represented graphically it will be seen that for small angles of incidence the reflection coefficient for horizontal polarisation is approximately -1. With vertical polarisation $|R|$ falls to a minimum at the Brewster angle and $\arg R$ changes from 180° to 0° . Appendix 4 gives some plots of the reflection coefficient for various types of reflecting surface.

The above formulae assume "smooth" reflecting surfaces. At the wavelengths under consideration most surfaces will appear smooth at small angles of incidence, but at larger angles diffraction effects will become apparent. These effects are not included in the propagation model.

2.2 Calculation of path loss.

The path loss between two antennas of gains g_T and g_R , at a frequency f and a distance d apart is given by,

$$L = 10 \log g_T + 10 \log g_R + 10 \log (c^2/4\pi) - 20 \log (f) + 10 \log (1/4\pi d^2)$$

The last term is the free space loss, the remaining terms take account of the antenna gains and the aperture of the receiving antenna which, of course, is frequency dependent.

Fig.2.1 illustrates the geometry of a typical multipath situation. If the contributions from the surface wave and any remote structures are ignored then the signal arriving at the receiving antenna is primarily composed of the direct ray and the ground reflected ray. The lengths of these paths are,

For the direct ray, $D = \sqrt{(h_1 - h_2)^2 + d^2}$

For the reflected ray, $R = \sqrt{(h_1 + h_2)^2 + d^2}$

and the grazing angle is,

$$\theta = \arctan \left(\frac{h_1 + h_2}{d} \right)$$




Fig.2.1 Multipath geometry.

The propagation model uses these relationships to work out the relative amplitudes and phases of the direct and reflected rays at the receiving antenna. Remember that the phase angle between the rays is a combination of the path length difference and the phase change due to the reflection coefficient. The phasor sum of the two rays is plotted as site attenuation versus distance where site attenuation is defined as,

$$\text{Site attenuation} = \frac{\text{Power transferred by direct connection}}{\text{Power transferred by propagation paths}}$$

3. Propagation prediction software.

A complete listing of the program is contained in Appendix 2. It is written in HP BASIC 5.1 and the COMPLEX binary must be resident. The user is prompted to enter frequency, range, antenna heights and gains, polarisation and the ground constants. Vertical scaling limits are set by the variables Ymax and Ymin and may be changed to suit a particular simulation.

4. Measurement of propagation loss.

4.1 Introduction.

The aim of the propagation measurement trial was primarily to validate the results produced by the propagation model. However, the choice of antenna types and the heights and distances involved was to a large extent determined by the application.

With reference to the definition of site attenuation given above it will be seen that a measurement of this quantity will require knowledge of the losses in the cables connecting the transmitter to the transmit antenna and the receive antenna to the receiver. This would be accomplished by connecting the cables directly together at the antennas and measuring the amount of power transferred at the frequency of interest. With the antennas reconnected and separated by a known distance a second measurement of transmitted power is made. The site attenuation is then the ratio of these two levels or in decibels,

$$\text{Site attenuation} = P_1 - P_2$$

where P_1 = power received by direct connection (dB)
 P_2 = power received via antennas (dB)

In practice, the measurement of P_1 need only be performed once at each frequency of measurement and used for any subsequent antenna separation and/or antenna type.

4.2 Description of equipment employed.

The transmit antennas were in a fixed position mounted on a tripod (except for the log-periodic antenna which was clamped onto a wooden stake driven into the ground). The receive antennas were mounted on a trolley which was moved by a motorised pulley arrangement up to a maximum distance of 50 metres. A serrated disc attached to one of the trolley wheels interrupted a light beam which caused pulses to be sent along a cable to a counter circuit inside the equipment cabin where the measurement software converted the pulses into distance travelled in metres.

For convenience a large proportion of the measurement procedure was carried out under computer control. A Hewlett-Packard 9000/310 machine was used with dual 3 1/2 inch disc drives for storage of programs and data. The RF measurements were made using an HP8753A Network Analyser which has a maximum output level of +25dBm and a 100dB dynamic range. Since single frequency measurements were being made the analyser was set to zero span which enabled readings to be taken fast enough for the required spatial resolution. In order to ensure a good signal to noise ratio a 1 Watt amplifier module was used at the output port of the network analyser. A listing of the software used to control the measurements is presented in Appendix 3.

The distances between the cabin housing the measuring equipment and the antennas necessitated long lengths of coaxial feeder cable. The loss of 100m of UR43 at 1GHz is about 70dB which would make meaningful measurements impossible.

Therefore Heliac cable was used where possible and UR67 where insufficient lengths of Heliac were available.

The following table lists the various combinations of antennas used in the trial.

<u>TRANSMIT ANTENNA</u>	<u>RECEIVE ANTENNA</u>	<u>FREQ. RANGE (MHz)</u>
Chase UHA 9105 (dipole, bicone)	Singer (dipole)	100,300,500,700,900
Chase UHA 9105 (dipole)	1/4 wave whip	400,600
CLP 5130-2 Log-periodic	CLP 5130-2 Log-periodic	100,300,500,700,900

The procedure for taking a typical measurement would be as follows;

- 1) Run measurement software.
- 2) Calibrate network analyser at the required frequency.
- 3) Measure cable losses at this frequency.
- 4) Set up required antennas.
- 5) Move trolley toward transmit antenna and stop at 2m distance.
- 6) Ensure distance counter has reset.
- 7) Start software taking readings.
- 8) Start trolley moving away from transmit antenna.
- 9) Stop trolley at 50m
- 10) Store data on disc.

4.3 Description of trial site.

All the propagation trials were carried out on the antennas field at RSRE South Site. This site was far from ideal being quite cluttered and not particularly flat. However, rather than use a more remote site with the attendant problems of transport, power and communication it was decided to tolerate the non-ideal conditions. Fig.4.1 shows a plan of the site.

The measuring equipment was housed in a shielded cabin with mains power taken from a nearby 15A power point.




Fig.4.1 Plan of trial site.

5. Results of the propagation measurement trial.

The results obtained from the trial are presented in a graphical format in Appendix 1. Each graph shows site attenuation in dB versus distance in metres both for the measured data and also the output of the prediction software. No measurement of the ground constants was undertaken and therefore typical values were used for the prediction. Small variations in the constants have little effect on the predicted results except to vary the sharpness of nulls.

The first set of measurements were intended to provide standard results against which the predictions of the propagation model could be compared. Half-wave dipoles were used for most measurements except at the lowest frequency (100 MHz) where a biconic antenna had to be used as the transmit antenna. Fig.A1.1 to Fig.A1.10 show the results of these measurements. In the prediction software the gains of the antennas were specified as 2.13dBi, the gain of a half-wave dipole relative to an isotrope.

The next set of measurements (Fig.A1.11 to Fig.A1.14) show the measured and predicted path losses between a dipole (1.7m) and a 1/4 wave vertically polarised whip close to ground level. In Fig.A1.11 the dipole was vertically polarised and the operating frequency was 400 MHz. In Fig.A1.12 the transmitting dipole was horizontally polarised and the whip vertically polarised but the simulation assumed both antennas horizontally polarised. This plot therefore gives some idea of the loss due to cross polarisation. Fig.A1.13 and Fig.A1.14 repeat these results but at 600 MHz.

The final measurements were performed using wide-band log-periodic antennas. Results were obtained every 200 MHz from 100 MHz to 900 MHz. The predicted results still assume dipole gains and so an indication of the gain of the log-periodic antennas can be obtained.

6. Discussion of trial results.

It can be seen by referring to the results in Appendix 1 that the measured and predicted results generally agree quite closely. However, there are some interesting discrepancies, most notably with regard to the position of the nulls. This is thought to be a consequence of the variation in height of the ground along which the trolley was moved, small changes in height can produce significant changes in null position. This error would also account for some of the discrepancy at longer ranges, for example in Fig.A1.1, if the nulls between 10m and 15m coincided then the error approaching 50m would not be so great. The position of the nulls is also affected by the angle of the ground reflection coefficient, but with horizontal polarisation the angle is almost always close to 180°. It is also possible that the surface wave changes null position but at frequencies greater than about 100 MHz the effect will be very small.

It is interesting to note that vertical polarisation suffers far less from deep nulls than horizontal. Appendix 4 shows that this is because the magnitude of the ground reflection coefficient is generally less with vertical polarisation than that with horizontal.

Discrepancies in the far field could in part be due to variation of the dipole gains with frequency. The predictions from the model assume constant gains of 2.13 dBi.

A further source of error will arise from the rather arbitrary choice of ground electrical parameters for the predicted results. Investigation of the effects of changing the ground parameters in the model has revealed that vertical polarisation at low frequencies is most sensitive in this respect. With horizontal polarisation, the most noticeable effect is on the depth of the nulls. These conclusions are in agreement with a similar investigation carried out in Ref.1.

Other possible sources of error are, reflections from nearby buildings and passing vehicles, signal cable leakage and additional radiation due to braid currents (imperfect balanced to unbalanced transformation).

7. Conclusions.

This report has presented the results of a comparison between measured propagation loss and predicted values from a simple computer model. Despite the simplicity of the model it has been shown to agree quite closely with the measured results. Furthermore, predictions from the model also tally with results from other models (Ref.1 and Ref.2).

It is intended to incorporate the model into other more complex simulations. This is relatively straightforward since the prediction program is quite short. Data generated by the package can be stored in arrays for further processing rather than being displayed directly.

8. Acknowledgements.

The author would like to thank A.W.Walker for his help in organising and running the propagation trial and especially with regard to setting up the trolley and motorised pulley system.

9. References.

1. EDWARDS, J.A., BAGWELL, D.J., MEHLER, M.J., THOMAS, D., CHECKETTS, D.G., "Short Range V.H.F. Propagation", RSRE Research Contract Ref. ML32b/1557, Final Report September 1987.
2. WILKINSON, Dr.T.A., MATTHEWS, Prof.P.A., "Radio Communication Through Vegetation", MoD Agreement Ref. D/ER1/9/4/2062/138. Annual Report 21 July 1989.
3. REED, H.R., RUSSELL, C.M., "Ultra High Frequency Propagation", Wiley, Chapman & Hall, 1953, Chapter 4, pp 82 - 99, Chapter 5, pp 117 - 135

APPENDIX 1.

Results of propagation loss measurements.

FIG.A1.1




FIG.A1.2




FIG.A1.3




FIG.A1.4




FIG.A1.5




FIG.A1.6




FIG. A1.7




FIG.A1.8




FIG.A1.9




FIG.A1.10




FIG.A1.11




FIG.A1.12




FIG.A1.13




FIG.A1.14




FIG.A1.15




FIG.A1.16




FIG.A1.17




FIG.A1.18





FIG.A1.19



APPENDIX 2.

Propagation prediction software

```

10 ! .....
20 !
30 !
40 !   SHORT RANGE VHF / UHF PROPAGATION
50 !   PREDICTION PROGRAM
60 !
70 !
80 !           P.R.Bellamy   May 1989
90 !           VER 2   August 1989
100!
110!   This software requires BASIC 5.1 with COMPLEX binary
120!
130! .....
140 COMPLEX N2,R,B,A
150 REAL Free_space,Fresnel,D,Resultant,Freq,Range,H1,H2,Er,Sigma,Inc
160 REAL Ymax,Ymin,L,C,EO,Lambda,Antenna_size,Antenna_factor,W,Topline,Nextlin
170 REAL Graze,Theta,Angle,Ref_coeff,P1,P2,X,Y
180 GINIT
190 KEY LABELS OFF
200 DEG
210 CLEAR SCREEN
220 PRINT "          SHORT RANGE PROPAGATION MODEL
230 INPUT "FREQUENCY (MHz)?",Freq
240 INPUT "MAX. RANGE (m)?",Range
250 INPUT "TX ANTENNA HEIGHT (m)?",H1
260 INPUT "RX ANTENNA HEIGHT (m)?",H2
270 INPUT "TX ANTENNA GAIN (dBi)?",Gt
280 INPUT "RX ANTENNA GAIN (dBi)?",Gr
290 INPUT "POLARISATION (H,V,Both)",Pol$
300 INPUT "RELATIVE PERMITTIVITY?",Er
310 INPUT "CONDUCTIVITY?",Sigma
320 LINPUT "OUTPUT TO CRT OR PLOTTER?",Dev$
330 !
340 IF Dev$(1)="-P" OR Dev$(1)="-p" THEN
350     LINPUT "PLOTTER ADDRESS?",Add$
360     PLOTTER IS VAL(Add$),"HPGL"
370 ELSE
380     PLOTTER IS CRT,"INTERNAL"
390 END IF
400 Inc=Range/300           ! 300 HORIZONTAL POINTS
410 !
420 !
430 !
440 Ymax=100
450 Ymin=0
460 VIEWPORT 10,120,7,85
470 PEN 1
480 WINDOW 0,Range,Ymax,Ymin
490 GRID Range/10,10,0,Ymax
500 CLIP OFF
510 CSIZE 3,.5
520 MOVE Range/2,Ymax+.07*(Ymax-Ymin)
530 LOG 5
540 LABEL "Range(m)"

```

```

550 FOR L=0 TO Range STEP Range/10
560   MOVE L,Ymax+.03*(Ymax-Ymin)
570   LABEL L
580 NEXT L
590 MOVE -.08*Range,Ymax-(Ymax-Ymin)/2
600 LDIR 90
610 LABEL "Site attenuation (dB)"
620 LDIR 0
630 FOR L=Ymin TO Ymax STEP 10
640   MOVE -.03*Range,L
650   LABEL L
660 NEXT L
670 !
680 !
690 !
700 !
710 !
720 C=3.E+8      !   SPEED OF LIGHT
730 EO=8.854E-12
740 Lambda=300/Freq
750 Antenna_size=Lambda/2!   ROUGH SIZE OF AERIAL
760 Fresnel=2*Antenna_size*Antenna_size/Lambda!   NEAR FIELD / FAR FIELD
770 W=2*PI*Freq*1.E+6
780 Antenna_factor=Gt+Gr+10*LGT(C*C/(4*PI))-20*LGT(Freq*1.E+6)
790 !
800 !
810 !
820 Topline=Ymin-.1*(Ymax-Ymin)
830 Nextline=Ymin-.05*(Ymax-Ymin)
840 LORG 2
850 MOVE 0,Topline
860 LABEL "Frequency: ";Freq;" MHz"
870 !
880 !
890 MOVE Range/2,Topline
900 IF Pol$="H" THEN LABEL "Horizontal"
910 IF Pol$="V" THEN LABEL "Vertical"
920 IF Pol$="B" THEN
930   PEN 2
940   MOVE Range/2,Topline
950   DRAW Range/1.8,Topline
960   PENUP
970   PEN 1
980   LABEL " Horizontal"
990   MOVE Range/1.4,Topline
1000  PEN 3
1010  DRAW Range/1.3,Topline
1020  PENUP
1030  PEN 1
1040  LABEL " Vertical"
1050 END IF
1060 !
1070 !
1080 MOVE 0,Nextline
1090 LABEL "TX :";H1;"m   RX :";H2;"m"
1100 MOVE Range/2,Nextline
1110 LABEL "Conductivity: ";Sigma;"   Er: ";Er
1120 !

```

```

1130 !
1140 !
1150 Next_pol:IF Pol$="V" THEN
1160     PEN 3
1170 ELSE
1180     PEN 2
1190 END IF
1200 !
1210 FOR D=Fresnel TO Range STEP Inc
1220     P1=SQR((H1-H2)*(H1-H2)+D*D) ! DIRECT PATH LENGTH
1230     P2=SQR((H1+H2)*(H1+H2)+D*D) ! REFLECTED PATH LENGTH
1240     Graze=ATN((H1+H2)/D) ! GRAZING ANGLE IN DEGREES
1250     Theta=360*((P2-P1)/Lambda-INT((P2-P1)/Lambda))! PATH DIFFERENCE IN
1260 ! DEGREES (i.e. PHASE)
1270 !
1280 !
1290 !*****
1300 ! CALCULATE REFLECTION COEFFICIENT
1310 !*****
1320     N2=CMPLX(Er,-Sigma/(W*EO))
1330     B=SQR(N2-COS(Graze)*COS(Graze))
1340     A=SIN(Graze)
1350     IF Pol$="V" THEN
1360         R=(N2*A-B)/(N2*A+B)
1370     ELSE
1380         R=(A-B)/(B+A)
1390     END IF
1400     Ref_coeff=ABS(R)
1410     Angle=ARG(R)
1420 !*****
1430 !
1440 !
1450 ! ADD PHASORS.....
1460     Free_space_d=SQR(30/(P1*P1))! FREE SPACE LOSS: DIRECT RAY
1470     Free_space_r=SQR(30/(P2*P2))! FREE SPACE LOSS: REFLECTED RAY
1480     X=Free_space_d+Free_space_r*Ref_coeff*COS(Theta+Angle)
1490     Y=Free_space_r*Ref_coeff*SIN(Theta+Angle)
1500     Resultant=(X*X+Y*Y)/(120*PI)
1510 !
1520 !
1530 !
1540 !*****
1550 ! PLOT THE RESULT...
1560     PLOT D,-10*LGT(Resultant)-Antenna_factor
1570 !*****
1580 !
1590 !
1600 NEXT D
1610 IF Pol$="B" THEN
1620     Pol$="V"
1630     PENUP
1640     GOTO Next_pol
1650 END IF
1660 !
1670 !
1680 PENUP
1690 !
1700 END

```

```

1710 !
1720 !
1730 !
1740 !
1750 !*****
1760 !          TYPICAL GROUND CONSTANTS
1770 !
1780 !          SIGMA (Mhos/M)          EPSILON
1790 !      Sea water                    5            80
1800 !      Fresh water                  8e-3          80
1810 !      Dry sandy coastal             2e-3          10
1820 !      Marshy forested               8e-3          12
1830 !      Rich agricultural             1e-2          15
1840 !      Pastoral rolling              5e-3          13
1850 !      Rocky, mountainous            2e-3           5
1860 !      Mountainous                   1e-3           5
1870 !      Cities, residential            2e-3           5
1880 !      Cities, industrial             1e-4           3
1890 !
1900 !
1910 !

```

APPENDIX 3.

Trial control software

```

10  !*****
20  !
30  !      S H O R T   R A N G E   P R O P A G A T I O N   T R I A L
40  !
50  !                      C O N T R O L   S O F T W A R E
51  !
52  !          (c) P.R.Bellamy / I.R.Dixon    1 Aug 1989
53  !
70  !*****
80  !
81 Start:  !
90  OPTION BASE 1
100 DEG
110 !*****
120 !                      V a r i a b l e   d e c l a r a t i o n s
130 !*****
140 DIM Message$(40),Answer$(1),File$(10),K$(1)
150 REAL Prop_data(32767,2)
160 INTEGER Big_cog_teeth,Lit_cog_teeth,Max_readings,Max_distance,Pulse_per_re
v,Net_an,Gpio,Readings
170 INTEGER Cal_done,True,False,Wanted_freq
180 REAL Freq,Dist_per_pulse,Gear_ratio,Wheel_circum,Max_dist_pulse,Null
190 !
200 !
210 !*****
220 !                      M A I N   P R O G R A M
230 !*****
231 !
240 GOSUB Initial_values
241 !
250 Input_freq(Freq)
251 !
260 Set_up_8753(Freq)
262 !
270 Calibrate_8753
271 !
280 Check_freq(Freq,Wanted_freq)
281 IF NOT Wanted_freq THEN
282     Entry_beep
284     DISP "CONTINUE WITH ";Freq;" MHz (y)?"
285     Wait_key(K$)
286     IF K$<>"Y" AND K$<>"y" THEN GOTO Start
287 END IF
288 !
290 Reset_counter(Gpio)
291 !
300 GOSUB Take_readings
301 !
303 !GOSUB Test_data
304 !
306 KEY LABELS OFF
307 REDIM Prop_data(Readings-1,2)
308 !
310 Plot_data(Prop_data(*),Dist_per_pulse)
311 !
312 DISP "Store data on disc? (n)"

```

```

313 Entry_beep
315 Wait_key(K$)
316 IF K$="N" OR K$="n" THEN
317     DISP "NO! ARE YOU ABSOLUTELY SURE?"
318     Entry_beep
319     Wait_key(K$)
321     IF K$="Y" OR K$="y" THEN
322         CLEAR SCREEN
323         PRINT "PROGRAM TERMINATED"
324         PAUSE
325     END IF
326 END IF
327 !
328 Store_data(Prop_data(*))
329 !
330 CLEAR SCREEN
331 DISP "FINISHED."
333 |*****
334 |*****
335 STOP
340 Initial_values: !
350 PRINTER IS CRT
360 CLEAR SCREEN
370 KEY LABELS OFF
380 PRINT CHR$(140);CHR$(132);"          S H O R T   R A N G E   P R O P A G A T
I O N   T R I A L          ";CHR$(136);CHR$(128)
390 PRINT "

          Initialising variables.....";
400 GINIT
410 Max_readings=5000          !Max number of readings (<32768)
420 Max_distance=100          !Distance in metres
430 !
440 Wheel_circum=.776
450 Big_cog_teeth=162
460 Lit_cog_teeth=20
470 Gear_ratio=Big_cog_teeth/Lit_cog_teeth
480 Pulse_per_rev=100
490 Dist_per_pulse=Wheel_circum/(Gear_ratio*Pulse_per_rev)
500 Max_dist_pulse=INT(Max_distance/Dist_per_pulse)-32768
510 !
520 Net_an=716
530 Gpio=12
540 !
550 White=1
560 Red=2
570 Yellow=3
580 Green=4
590 Cyan=5
600 Blue=6
610 Magenta=7
620 !
630 PRINT "done."
640 RETURN
650 !
660 !
670 Take_readings: !
680 CLEAR SCREEN

```

```

681 PRINT "          DATA GATHERING ROUTINE"
682 KEY LABELS ON
683 CONTROL 2,2;1
684 ON KEY 1 LABEL " ABORT" GOTO Abort
686 PRINT "

```

PRESS KEY 1 TO ABORT MEASUREMENT"

```

687 !
688 Readings=1
689 Entry_beep
691 DISP "PRESS RETURN TO START MEASUREMENT..."
692 Wait_key(K$)
693 DISP "TAKING DATA....."
694 REPEAT
695     OUTPUT Net_an;"OUTPMARK;"
696     ENTER Net_an;Prop_data(Readings,2),Null,Null
697     CONTROL Gpio,2;0
698     STATUS Gpio,3;Prop_data(Readings,1)
699     CONTROL Gpio,2;1
700     Readings=Readings+1
701 UNTIL Readings>Max_readings
702 Err(3)
703 RETURN
704 Abort: !
705 Err(2)
706 RETURN
920 !
921 !
922 Test_data: !
930 CLEAR SCREEN
931 PRINT "          DATA GATHERING ROUTINE"
932 PRINT "

```

WARNING: THIS ROUTINE DOES NOT READ THE DISTANCE COUNTER"

```

934 KEY LABELS ON
935 CONTROL 2,2;1
936 ON KEY 1 LABEL " ABORT" GOTO Abort2
937 PRINT "

```

PRESS KEY 1 TO ABORT MEASUREMENT"

```

938 !
939 Readings=1
940 Entry_beep
942 DISP "PRESS RETURN TO START MEASUREMENT..."
943 Wait_key(K$)
944 DISP "TAKING DATA....."
945 TO-TIMEDATE
946 REPEAT
947     OUTPUT Net_an;"OUTPMARK;"
948     ENTER Net_an;Prop_data(Readings,2),Null,Null
949     Prop_data(Readings,1)=Readings*30.0
950     Readings=Readings+1
951 UNTIL Readings>Max_readings
952 Err(3)

```

```

953 RETURN
954 Abort2: !
955 T1-TIMEDATE
956 PRINT "TIME PER READING:";(T1-T0)/Readings;" SECS"
957 Err(2)
958 RETURN
959 !
960 !
961 !
962 END
963 !=====
964 ! SUB PROGRAM DEFINITIONS
965 Set_up_8753: !
966 SUB Set_up_8753(REAL Freq)
967 Na=716
968 ABORT 7
969 OUTPUT Na;"PRES;" !PRESET 8753
970 OUTPUT Na;"FORM3;POIN 3;" !FORMAT 3, 3 POINTS
971 OUTPUT Na;"LOGM;" !LOG MAG DISPLAY
972 OUTPUT Na;"CHAN1;" !CHANNEL 1 ACTIVE
975 OUTPUT Na;"IFBW30;" !IF BANDWIDTH
976 OUTPUT Na;"POWE 0;" !SOURCE POWER
977 OUTPUT Na;"S21;" !TRANSMISSION MEASUREMENT
978 OUTPUT Na;"CENT ";Freq;" MHZ;SPAN 0 MHZ;" !SET CENTRE FREQ AND SPAN
979 !
980 A$=CHR$(34)&"PROPCAL"&CHR$(34) !TITLE REG 5
981 OUTPUT Na;"TITR5;"&A$&";"
982 SUBEND
983 !
984 !
985 Calibrate_8753: !
986 SUB Calibrate_8753
987 CLEAR SCREEN
988 PRINT "
989 PRINT " 8 7 5 3 CALIBRATION "

```

```

"
990 Entry_beep
991 INPUT "LOAD CAL DATA FROM 8753 REG 5? (Y/N)".Reply$
992 IF Reply$="Y" OR Reply$="y" THEN
993 OUTPUT 716;"OPC?;RECA5;"
994 ENTER 716;Reply
995 ELSE
996 OUTPUT 716;"CALKN50;"
997 OUTPUT 716;"CALIFUL2;" !FULL 2 PORT CALIBRATION
998 OUTPUT 716;"REFL;"
999 !
1000 DISP "CONNECT OPEN TO PORT 1 THEN PRESS RETURN..."
1001 CALL Wait_key(K$)
1002 OUTPUT 716;"CLASS11A;OPC?;STANB;"
1003 ENTER 716;Reply
1004 !
1005 DISP "CONNECT SHORT TO PORT 1 THEN PRESS RETURN..."

```

```

1006     CALL Wait_key(K$)
1007     OUTPUT 716;"CLASS11B;OPC?;STANB;"
1008     ENTER 716;Reply
1009     !
1010     DISP "CONNECT LOAD TO PORT 1 THEN PRESS RETURN..."
1011     CALL Wait_key(K$)
1012     OUTPUT 716;"CLASS11C;OPC?;STANA;"
1013     ENTER 716;Reply
1014     !
1015     DISP "CONNECT OPEN TO PORT 2 THEN PRESS RETURN..."
1016     CALL Wait_key(K$)
1017     OUTPUT 716;"CLASS22A;OPC?;STANB;"
1018     ENTER 716;Reply
1019     !
1020     DISP "CONNECT SHORT TO PORT 2 THEN PRESS RETURN..."
1021     CALL Wait_key(K$)
1022     OUTPUT 716;"CLASS22B;OPC?;STANB;"
1023     ENTER 716;Reply
1024     !
1025     DISP "CONNECT LOAD TO PORT 2 THEN PRESS RETURN..."
1026     CALL Wait_key(K$)
1027     OUTPUT 716;"CLASS22C;OPC?;STANA;"
1028     ENTER 716;Reply
1029     OUTPUT 716;"REFD;"
1030     !
1031     !
1032     DISP "CONNECT THRU THEN PRESS RETURN..."
1033     CALL Wait_key(K$)
1034     OUTPUT 716;"TRAN;"
1035     !
1036     OUTPUT 716;"OPC?;FWDI;"
1037     ENTER 716;Reply
1038     !
1039     OUTPUT 716;"OPC?;REVT;"
1040     ENTER 716;Reply
1041     !
1042     OUTPUT 716;"OPC?;FWDI;"
1043     ENTER 716;Reply
1044     !
1045     OUTPUT 716;"OPC?;REVM;"
1046     ENTER 716;Reply
1047     OUTPUT 716;"TRAD;"
1048     !
1049     OUTPUT 716;"ISOL;"
1050     OUTPUT 716;"OMII;"
1051     OUTPUT 716;"ISOD;"
1052     !
1053     OUTPUT 716;"SAV2;"
1054     OUTPUT 716;"OPC?;SAVE5;"
1055     ENTER 716;Reply
1056     Cal_done=True
1057     END IF
1058     DISP "CALIBRATION COMPLETE"
1059     WAIT 1
1060     CLEAR SCREEN
1061     SUBEND
1062     !
1063     !

```

```

1064 Wait_key: !
1065 SUB Wait_key(K$)
1066   DIM T$[10]
1067   ON KBD GOTO Quit
1068   LOOP
1069   END LOOP
1070 Quit: !
1071   T$=KBD$
1072   K$=T$[1,1]
1073 SUBEND
1074 !
1075 !
1076 Input_freq: !
1077 SUB Input_freq(REAL Freq)
1078   Entry_beep
1079   LOOP
1080     INPUT "Frequency for this measurement (MHz) ",Freq
1081     EXIT IF NOT (Freq<100) AND NOT (Freq>1000)
1082     Err(1)
1083   END LOOP
1084 SUBEND
1085 !
1086 !
1087 Err: !   MASTER ERROR HANDLING ROUTINE
1088 SUB Err(INTEGER Err_num)
1089   SELECT Err_num
1090   !
1091   CASE 1
1092     DISP "FREQUENCY OUT OF RANGE";
1093   !
1094   CASE 2
1095     DISP "MEASUREMENT ABORTED";
1096   !
1097   CASE 3
1098     DISP "MAXIMUM NUMBER OF READINGS EXCEEDED";
1099   !
1100   CASE 4
1101     DISP "CALIBRATION DATA NOT VALID AT THIS FREQUENCY";
1102   !
1103   CASE 5
1104     DISP "RESET FAILED"
1105   !
1106   CASE ELSE
1107     END SELECT
1108     DISP " <SPACE> TO CONT"
1109     BEEP 1000,.2
1110     Wait_key(K$)
1111     DISP "
1112   SUBEND
1113 !
1114 !
1115 !
1116 Entry_beep: !
1117 SUB Entry_beep
1118   BEEP 900,.05
1119 SUBEND
1120 !
1121 !
1122 Reset_counter: !

```

```

1123 SUB Reset_counter(INTEGER Gpio)
1124     INTEGER A
1125     ! Outputs a reset pulse to the distance counter on CTL1
1126     Do_again: !
1127     CONTROL Gpio,2;1! CTL1=1 (RESET HIGH), CTLO=0 (CONTROL LOW)
1128     WAIT .1
1129     CONTROL Gpio,2;3! CTL1=0 (RESET LOW ), CTLO=0 (CONTROL LOW)
1130     WAIT .1
1131     CONTROL Gpio,2;1! CTL1=1 (RESET HIGH), CTLO=0 (CONTROL LOW)
1132     !
1133     DISP "RESET PULSE SENT - MOVE WHEEL THEN PRESS A KEY.."
1134     Wait_key(A$)
1135     CONTROL Gpio,2;0
1136     STATUS Gpio,3;A
1137     CONTROL Gpio,2;1
1138     IF A<0 OR A>100 THEN
1139         Err(5)
1140         GOTO Do_again
1141     END IF
1142 SUBEND
1143 !
1144 !
1145 Check_freq: !
1146 SUB Check_freq(REAL Freq,INTEGER Flag)
1147     REAL F
1148     Flag=1
1149     OUTPUT 716;"CENT?;"
1150     ENTER 716;F
1151     IF F<>Freq*1.E+6 THEN
1152         CALL Err(4)
1153         Freq=F/1.E+6
1154         Flag=0
1155     END IF
1156 SUBEND
1157 !
1158 !
1159 Plot_data: !
1160 SUB Plot_data(REAL Prop_data(*),Dist_per_pulse)
1161     INTEGER White,Red,Yellow,Green,Cyan,Blue,Magenta
1162     INTEGER I
1163     White=1
1164     Red=2
1165     Yellow=3
1166     Green=4
1167     Cyan=5
1168     Blue=6
1169     Magenta=7
1170 !
1171     CLEAR SCREEN
1172     GINIT
1173     GCLEAR
1174     GRAPHICS ON
1175     INPUT "Start distance: ",Start_pos
1176     INPUT "Amp gain (dB): ",Gain
1177     PEN White
1178     Readings=SIZE(Prop_data,1)
1179 !
1180 VIEWPORT 10,120,28,95

```

```

1183 WINDOW 0,50,100,0
1184 AREA PEN Blue
1185 MOVE 0,0
1186 RECTANGLE 50,100,FILL
1187 AXES 1,2,0,100,10,5,3
1188 !
1189 CLIP OFF
1190 CSIZE 2.5
1191 LDIR 0
1192 LORG 8
1193 FOR I=0 TO 100 STEP 10
1194     MOVE 0,I
1195     LABEL I
1196 NEXT I
1197 CSIZE 3
1198 LORG 4
1199 LDIR 90
1200 MOVE -3,50
1201 LABEL "Site attenuation (dB)"
1202 !
1203 CSIZE 2.5
1204 LDIR 0
1205 LORG 6
1206 FOR I=0 TO 50 STEP 10
1207     MOVE I,100
1208     LABEL I
1209 NEXT I
1210 CSIZE 3
1211 MOVE 25,102
1212 LABEL "Range (m)"
1213 !
1214 MOVE 40,0
1215 LDIR 0
1216 LORG 4
1217 CSIZE 2.7
1218 LABEL DATE$(TIMEDATE); " "; TIME$(TIMEDATE)
1219 !
1220 FOR I=2 TO Readings
1221     Dist=FNDistance(Prop_data(I,1),Dist_per_pulse)+Start_pos
1222     Prop_data(I,2)=Prop_data(I,2)+Gain
1223     PLOT Dist,Prop_data(I,2)
1224     Prop_data(I,1)=Dist
1225 NEXT I
1226 !
1227 SUBEND
1228 !
1229 !
1230 DEF FNDistance(REAL Counter,Dist_per_pulse)
1231     IF Counter<0 THEN
1232         RETURN (Counter+65536)*Dist_per_pulse
1233     ELSE
1234         RETURN Counter*Dist_per_pulse
1235     END IF
1236 FNEND
1237 !
1238 !
1239 Store_data: !
1240     ! Creates file on which the data in Prop_data is stored

```



```

1241      ! Summary$ is a 160 character (2 line) summary
1242 SUB Store_data(REAL Prop_data(*))
1243   ON ERROR GOTO Trap
1244   CLEAR SCREEN
1245   DIM Summary$(160)
1246   DIM File$(25)
1247   Readings=SIZE(Prop_data,1)
1248   !
1249   !
1250   PRINT TABXY(1,21);"SUMMARY: (2 lines max)"
1251   LINPUT Summary$
1252 Inp:LINPUT "File:mass storage unit?",File$
1253   DISP "Creating file..."
1254   CREATE BDAT File$,Readings*16+500,1      ! NOTE: BASIC 5 seems to w
ork out      ! the correct file size itself!
1255
1256   ASSIGN @Path TO File$
1257   DISP "Storing data..."
1258   OUTPUT @Path;Summary$,TIMEDATE,Readings,Prop_data(*)
1259   ASSIGN @Path TO *
1260   DISP "
1261   SUBEXIT
1262   !
1263 Trap: !
1264   DISP ERRMS;" <SPACE> TO CONT"
1265   BEEP 1000,.2
1266   Wait_key(K$)
1267   GOTO Inp
1268 SUBEND

```

APPENDIX 4.

Ground reflection coefficient plots





Frequency: 1.E+8 Hz

SEA WATER

Relative permittivity: 80

Conductivity: 5 mhos/m





Frequency: 1.E+8 Hz

SEA WATER

Relative permittivity: 80

Conductivity: 5 mhos/m





Frequency: 1.E+8 Hz

MARSHY FORESTED

Relative permittivity: 12

Conductivity: .008 mhos/m





Frequency: 1.E+8 Hz

MARSHY FORESTED

Relative permittivity: 12

Conductivity: .008 mhos/m





Frequency: 1.E+8 Hz

URBAN RESIDENTIAL

Relative permittivity: 5

Conductivity: .002 mhos/m



Frequency: 1.E+8 Hz

URBAN RESIDENTIAL

Relative permittivity: 5

Conductivity: .002 mhos/m

DOCUMENT CONTROL SHEET

Overall security classification of sheet UNCLASSIFIED

(As far as possible this sheet should contain only unclassified information. If it is necessary to enter classified information, the box concerned must be marked to indicate the classification eg (R) (C) or (S))

1. DRIC Reference (if known)	2. Originator's Reference MEMO 4335	3. Agency Reference	4. Report Security U/C Classification	
5. Originator's Code (if known) 7784000	6. Originator (Corporate Author) Name and Location ROYAL SIGNALS AND RADAR ESTABLISHMENT ST ANDREWS ROAD, GREAT MALVERN, WORCESTERSHIRE WR14 3PS			
5a. Sponsoring Agency's Code (if known)	6a. Sponsoring Agency (Contract Authority) Name and Location			
7. Title SHORT RANGE, CLOSE TO GROUND, VHF/UHF PROPAGATION:				
7a. Title in Foreign Language (in the case of translations)				
7b. Presented at (for conference papers) Title, place and date of conference				
8. Author 1 Surname, initials BELLAMY P R	9(a) Author 2	9(b) Authors 3,4...	10. Date 1989.10	pp. ref. 47
11. Contract Number	12. Period	13. Project	14. Other Reference	
15. Distribution statement UNLIMITED				
Descriptors (or keywords)				
continue on separate piece of paper				
Abstract This report describes a VHF/UHF propagation prediction model and a series of field measurements aimed at verifying the results of the model. Comparisons between the model and the trial results were made at spot frequencies from 100 MHz to 900 MHz for several types of antenna and for various heights (not exceeding 2 metres) above ground level. The predicted path losses from the model were found to agree quite well with the results of the field measurements. It should be noted that the prediction model was never intended to be a comprehensive tool, merely a utility package which could be incorporated into other software. Continued of next page				

CONTINUED

The program was written on an HP 9000 series model 310 computer using BASIC 5.1. This language has a set of commands which allow direct manipulation of complex values, consequently, translation into another language or version of BASIC not having these facilities will require extra programming.

Great Britain. 1/10/68